<?xml version="1.0" encoding="UTF-8"?>

<xsl:stylesheet version="1.0" xmlns:xsl="http://www.w3.org/1999/XSL/Transform">
	
	<xsl:include href="../admin.xsl" />
	<xsl:include href="../common/form.xsl" />
	<xsl:include href="../cp/path.xsl" />
	
	<!-- Запускаем рендер страницы -->
	<xsl:template match="/data">
		<html>
		
		<head>
			<title>
				<xsl:value-of select="$pageTitle" />
			</title>
			<xsl:call-template name="meta-head" />
			<link rel="stylesheet" href="{$baseURI}/css/admin/forms.css" type="text/css" media="all" />
		</head>
		<body>
			<xsl:call-template name="admin-header" />
			
			<div id="blocks">
				<div id="blocks_inner" />
			</div>
			
			<div id="content">
				<div id="content_inner">
					<xsl:apply-templates select="response" />
				</div>
			</div>
			
			<xsl:call-template name="footer" />
		</body>
		</html>
	</xsl:template>
	
	<xsl:template match="response">
		<div id="path">
			<xsl:apply-templates select="document('xmodule://cp/getPath/0;banners')/data/response" mode="cp-path">
				<xsl:with-param name="activeLink">1</xsl:with-param>
			</xsl:apply-templates>
		</div>
		
		<h1>
			<xsl:if test="$method = 'add_banner'">
				<xsl:text>Новый баннер</xsl:text>
			</xsl:if>
			<xsl:if test="$method = 'edit_banner'">
				<xsl:text>Редактирование баннера</xsl:text>
			</xsl:if>
		</h1>
		
		<form action="{$baseURI}/admin/banners/add_banner/save" method="post">
			<xsl:if test="$method = 'edit_banner'">
				<xsl:attribute name="action">
					<xsl:value-of select="concat($baseURI, '/admin/banners/edit_banner/', $formPrefix, '/save')" />
				</xsl:attribute>
			</xsl:if>
			
			<xsl:apply-templates select="form" />
			
			<p>
				<input type="submit" value="Добавить баннер">
					<xsl:if test="$method = 'edit_banner'">
						<xsl:attribute name="value">
							<xsl:text>Сохранить</xsl:text>
						</xsl:attribute>
					</xsl:if>
				</input>
			</p>
		</form>
	</xsl:template>
	
</xsl:stylesheet>